

# COMPUTER-SUPPORTED ANALYSIS OF POSITIVE PROPERTIES, ULTRAFILTERS AND MODAL COLLAPSE IN VARIANTS OF GÖDEL'S ONTOLOGICAL ARGUMENT

## Abstract

Three variants of Kurt Gödel's ontological argument, proposed by Dana Scott, C. Anthony Anderson and Melvin Fitting, are encoded and rigorously assessed on the computer. In contrast to Scott's version of Gödel's argument the two variants contributed by Anderson and Fitting avoid modal collapse. Although they appear quite different on a cursory reading they are in fact closely related. This has been revealed in the computer-supported formal analysis presented in this article. Key to our formal analysis is the utilization of suitably adapted notions of (modal) ultrafilters, and a careful distinction between extensions and intensions of positive properties.

*Keywords:* Computational metaphysics, ontological argument, higher-order modal logic, higher-order logic, automated reasoning, modal ultrafilters.

## 1. Introduction

The premises of the variant of the modal ontological argument [20] which was found in Kurt Gödel's "Nachlass" are inconsistent; this holds already in base modal logic **K** [11, 9]. The premises of Scott's [28] variant of Gödel's work, in contrast, are consistent [9, 11], but they imply the modal collapse,  $\varphi \rightarrow \Box\varphi$ , which has by many philosophers been considered an undesirable side effect; cf. Sobel [30] and the references therein.<sup>1</sup>

---

<sup>1</sup>The modal collapse was already noted by Sobel [29, 30]. One might conclude from it, that the premises of Gödel's argument imply that everything is determined, or alternatively, that there is no free will. Srećko Kovač [25] argues that modal collapse was eventually intended by Gödel.

In this article we formally encode and analyze, starting with Scott’s variant, two prominent further emendations of Gödel’s work both of which successfully avoid modal collapse. These two variants have been contributed by C. Anthony Anderson [1, 2] and Melvin Fitting [16], and on a cursory reading they appear quite different. Our formal analysis, however, shows that from a certain mathematical perspective they are in fact closely related.

Two notions are particularly important in our analysis. From set theory, resp. topology, we borrow and suitably adapt, for use in our modal logic context, the notion of ultrafilter and apply it in two different versions to the set of positive properties. From the philosophy of language we adopt the distinction between intensions and extensions of (positive) properties. Such a distinction has been suggested already by Fitting in his book “*Types, Tableaus and Gödel’s God*” [16], which we take as a starting point in our formalization work.

Utilizing these notions, and extending Fitting’s analysis, the modifications as introduced by Anderson and Fitting to Gödel’s concept of positive properties are formally studied and compared. Our computer-supported analysis, which is carried out in the proof assistant system Isabelle/HOL [27], is technically enabled by the universal logical reasoning approach [4], which exploits shallow semantical embeddings (SSEs) of various logics of interest – such as intensional higher-order modal logics (IHOML) in the present article – in Church’s simple type theory [5], aka. classical higher-order logic (HOL). This approach enables the reuse of existing, interactive and automated, theorem proving technology for HOL to mechanize also non-classical higher-order reasoning.

Some of the findings reported in this article have, at an abstract level, already been summarized in the literature before [24, 6, 17], but they have not been published in full detail yet (for example, the notions of “modal” ultrafilters, as employed in our analysis, have not been made precise in these papers). This is the contribution of this article.

In fact, we present and explain in detail the SSE of intensional higher-order modal logic (IHOML) in HOL (§3.1), the encoding of different types of modal filters and modal ultrafilters in HOL (§3.2), and finally the encoding and analysis of the three mentioned variants of Gödel’s ontological argument in HOL utilizing the SSE approach (§4, §5 and §6). We start out (§2) by pointing to related prior work and by outlining the SSE approach.

## 2. Prior Work and the SSE Approach

The key ideas of the shallow semantical embedding (SSE) approach, as relevant for the remainder of this article, are briefly outlined. This section is intended to make the article sufficiently self-contained and to give references to related prior work. The presentation in this section is taken and adapted from a recently published related article [24, §1.1]; readers already familiar with the SSE approach may simply skip it, and those who need further details may consult further related documents [7, 4].

Earlier papers, cf. [4] and the references therein, focused on the development of SSEs. These papers show that the standard translation from propositional modal logic to first-order logic can be concisely modeled (i.e., embedded) within higher-order theorem provers, so that the modal operator  $\Box$ , for example, can be explicitly defined by the  $\lambda$ -term  $\lambda\varphi.\lambda w.\forall v.(Rwv \rightarrow \varphi v)$ , where  $R$  denotes the accessibility relation associated with  $\Box$ . Then one can construct first-order formulas involving  $\Box\varphi$  and use them to represent and proof theorems. Thus, in an SSE, the target logic is internally represented using higher-order constructs in a proof assistant system such as Isabelle/HOL. The first author, in collaboration with Paulson [7], developed an SSE that captures quantified extensions of modal logic (and other non-classical logics). For example, if  $\forall x.\phi x$  is shorthand in higher-order logic (HOL) for  $\Pi(\lambda x.\phi x)$ , then  $\Box\forall xPx$  would be represented as  $\Box\Pi'(\lambda x.\lambda w.Pxw)$ , where  $\Pi'$  stands for the  $\lambda$ -term  $\lambda\Phi.\lambda w.\Pi(\lambda x.\Phi xw)$ , and the  $\Box$  gets resolved as described above.

To see how these expressions can be resolved to produce the right representation, consider the following series of reductions:

$$\begin{aligned}
\Box\forall xPx &\equiv \Box\Pi'(\lambda x.\lambda w.Pxw) \\
&\equiv \Box((\lambda\Phi.\lambda w.\Pi(\lambda x.\Phi xw))(\lambda x.\lambda w.Pxw)) \\
&\equiv \Box(\lambda w.\Pi(\lambda x.(\lambda x.\lambda w.Pxw)xw)) \\
&\equiv \Box(\lambda w.\Pi(\lambda x.Pxw)) \\
&\equiv (\lambda\varphi.\lambda w.\forall v.(Rwv \rightarrow \varphi v))(\lambda w.\Pi(\lambda x.Pxw)) \\
&\equiv (\lambda\varphi.\lambda w.\Pi(\lambda v.Rwv \rightarrow \varphi v))(\lambda w.\Pi(\lambda x.Pxw)) \\
&\equiv (\lambda w.\Pi(\lambda v.Rwv \rightarrow (\lambda w.\Pi(\lambda x.Pxw)v))) \\
&\equiv (\lambda w.\Pi(\lambda v.Rwv \rightarrow \Pi(\lambda x.Pxv))) \\
&\equiv (\lambda w.\forall v.Rwv \rightarrow \forall x.Pxv) \\
&\equiv (\lambda w.\forall vx.Rwv \rightarrow Pxv)
\end{aligned}$$

Thus, we end up with a representation of  $\Box\forall xPx$  in HOL. Of course, types are assigned to each term of the HOL language. More precisely, in the

SSE presented in Fig. 1, we will assign individual terms (such as variable  $x$  above) the type  $\mathbf{e}$ , and terms denoting worlds (such as variable  $w$  above) the type  $\mathbf{i}$ . From such base choices, all other types in the above presentation can be inferred. While types have been omitted above, they will often be given in the remainder of this article.

The SSE technique provided a fruitful starting point for a natural encoding of Gödel’s ontological argument in second-order modal logics **S5** and **KB** [9]. Initial studies investigated Gödel’s and Scott’s variants of the argument within the higher-order automated theorem prover (henceforth ATP) LEO-II [8]. Subsequent work deepened these assessment studies [11, 12]. Instead of using LEO-II, these studies utilized the higher-order proof assistant Isabelle/HOL, which is interactive and which also supports strong proof automation. Some experiments were also conducted with the proof assistant Coq [10]. Further work (see the references in [24, 4]) contributed a range of similar studies on variants of the modal ontological argument that have been proposed by Anderson [1], Anderson and Gettings [2], Hájek [21, 22, 23], Fitting [16], and Lowe [26]. Particularly relevant for this article is some prior formalization work by the authors that has been presented in [18, 17]. The use of ultrafilters to study the distinction between extensional and intensional positive properties in the variants of Scott, Anderson and Fitting has first been mentioned in invited keynotes presented at the AISSQ-2018 [6] and the FMSPH-2019 [3] conferences.

### 3. Further Preliminaries

The formal analysis in this article takes Fitting’s book [16] as a starting point; see also [18, 17]. Fitting suggests to carefully distinguish between intensions and extensions of positive properties in the context of Gödel’s argument, and, in order to do so within a single framework, he introduces a sufficiently expressive higher-order modal logic enhanced with means for the explicit representation of intensional terms and their extensions, which we have termed intensional higher-order modal logic (IHOML) in previous work [17]. The SSE of IHOML in HOL, that we utilize in the remainder of this article, is presented in §3.1. Notions of ultrafilters on sets of intensions, resp. extensions, of (positive) properties are then introduced in §3.2. Since we develop, explain and discuss our formal encodings directly in Isabelle/HOL [27], some familiarity with this proof assistant and its background logic HOL [5] is assumed.

### 3.1. Intensional Higher-Order Modal Logic in HOL

An encoding of IHOML in Isabelle/HOL utilizing the SSE approach, is presented in Fig. 1. It starts in line 3 with the declaration of two base types in HOL as mentioned before: type **i** stands for possible worlds and type **e** for entities/individuals. To keep the encoding concise some type synonyms are introduced in lines 4–7, which we explain next.

$\delta$  and  $\sigma$  abbreviate the types of predicates  $\mathbf{e} \Rightarrow \mathbf{bool}$  and  $\mathbf{i} \Rightarrow \mathbf{bool}$ , respectively. Terms of type  $\delta$  represent (extensional) properties of individuals. Terms of type  $\sigma$  can be seen to represent world-lifted propositions, i.e., truth-sets in Kripke’s modal relational semantics [19]. Note that the explicit transition from modal propositions to terms (truth-sets) of type  $\sigma$  is a key aspect in SSE approach; see the literature [4] for further details. In the remainder of this article we make use of phrases such as “world-lifted” or “ $\sigma$ -type” terms to emphasize this conversion in the SSE approach.

$\tau$ , which abbreviates the type  $\mathbf{i} \Rightarrow \mathbf{i} \Rightarrow \mathbf{bool}$ , stands for the type of accessibility relations in modal relational semantics, and  $\gamma$ , which stands for  $\mathbf{e} \Rightarrow \sigma$ , is the type of world-lifted, intensional properties.

In lines 8–32 in Fig. 1 the modal logic connectives are introduced. For example, in line 15 we find the definition of the world-lifted  $\vee$ -connective (which is of type  $\sigma \Rightarrow \sigma \Rightarrow \sigma$ ; type information is given here explicitly after the `::`-token for ‘`mor`’, which is the ASCII-denominator for the infix-operator  $\vee$  as introduced in parenthesis shortly after).  $\varphi_\sigma \vee \psi_\sigma$  is defined as abbreviation for the truth-set  $\lambda w_i. \varphi_\sigma w_i \vee \psi_\sigma w_i$  (i.e.,  $\vee$  is associated with the lambda-term  $\lambda \varphi_\sigma. \lambda \psi_\sigma. \lambda w_i. \varphi_\sigma w_i \vee \psi_\sigma w_i$ ). In the remainder we generally use bold-face symbols for world-lifted connectives (such as  $\vee$ ) in order to rigorously distinguish them from their ordinary counterparts (such as  $\vee$ ) in the meta-logic HOL.

The world-lifted  $\neg$ -connective is introduced in line 11,  $\perp$  and  $\top$  in lines 9–10, and respective further abbreviations for conjunction, implication and equivalence are given in lines 14, 16 and 17, respectively. The operators  $\neg$  and  $\rightarrow$ , introduced in lines 12 and 13, negate properties of types  $\delta$  and  $\gamma$ , respectively; these operations occur in the premises in the works of Scott, Anderson and Fitting which govern the definition of positive properties.

As we see in Fig. 1, types can often be omitted in Isabelle/HOL due to the system’s internal type inference mechanism. This feature is exploited in our formalization to some extent to improve readability. However, for all *new* abbreviations and definitions, we always explicitly declare the types

```

1 theory IHOML imports Main
2 begin
3   typedecl i (*Possible worlds*) typedecl e (*Individuals*)
4   type_synonym  $\delta$  = "e $\Rightarrow$ bool" (*Type of ordinary predicates*)
5   type_synonym  $\sigma$  = "i $\Rightarrow$ bool" (*Type of world-lifted propositions, truth-sets*)
6   type_synonym  $\tau$  = "i $\Rightarrow$ i $\Rightarrow$ bool" (*Type of accessibility relations*)
7   type_synonym  $\gamma$  = "e $\Rightarrow$  $\sigma$ " (*Type of lifted predicates*)
8 (**Logical operators lifted to truth-sets**)
9   abbreviation mbot::" $\sigma$ " ("⊥") where "⊥  $\equiv$   $\lambda$ w. False"
10  abbreviation mtop::" $\sigma$ " ("⊤") where "⊤  $\equiv$   $\lambda$ w. True"
11  abbreviation mnot::" $\sigma \Rightarrow \sigma$ " ("¬" [52]53) where "¬ $\varphi$   $\equiv$   $\lambda$ w. ¬( $\varphi$  w)"
12  abbreviation negpred::" $\delta \Rightarrow \delta$ " ("¬" [52]53) where "¬ $\Phi$   $\equiv$   $\lambda$ x. ¬( $\Phi$  x)"
13  abbreviation mnegpred::" $\gamma \Rightarrow \gamma$ " ("¬" [52]53) where "¬ $\Phi$   $\equiv$   $\lambda$ x.  $\lambda$ w. ¬( $\Phi$  x w)"
14  abbreviation mand::" $\sigma \Rightarrow \sigma \Rightarrow \sigma$ " (infixr"∧"51) where " $\varphi$  ∧  $\psi$   $\equiv$   $\lambda$ w. ( $\varphi$  w) ∧ ( $\psi$  w)"
15  abbreviation mor::" $\sigma \Rightarrow \sigma \Rightarrow \sigma$ " (infixr"∨"50) where " $\varphi$  ∨  $\psi$   $\equiv$   $\lambda$ w. ( $\varphi$  w) ∨ ( $\psi$  w)"
16  abbreviation mimp::" $\sigma \Rightarrow \sigma \Rightarrow \sigma$ " (infixr"→"49) where " $\varphi$  →  $\psi$   $\equiv$   $\lambda$ w. ( $\varphi$  w) → ( $\psi$  w)"
17  abbreviation mequ::" $\sigma \Rightarrow \sigma \Rightarrow \sigma$ " (infixr"↔"48) where " $\varphi$  ↔  $\psi$   $\equiv$   $\lambda$ w. ( $\varphi$  w) ↔ ( $\psi$  w)"
18 (**Polymorphic possibilist quantification**)
19  abbreviation mforall::"( $\delta \Rightarrow \sigma$ )  $\Rightarrow$   $\sigma$ " ("∀" [8]9) where "∀ $\Phi$   $\equiv$   $\lambda$ w.  $\forall$ x. ( $\Phi$  x w)"
20  abbreviation mforallB (binder"∀"[8]9) where "∀x.  $\varphi$ (x)  $\equiv$  ∀ $\varphi$ "
21  abbreviation mexists::"( $\delta \Rightarrow \sigma$ )  $\Rightarrow$   $\sigma$ " ("∃" [8]9) where "∃ $\Phi$   $\equiv$   $\lambda$ w.  $\exists$ x. ( $\Phi$  x w)"
22  abbreviation mexistsB (binder"∃"[8]9) where "∃x.  $\varphi$ (x)  $\equiv$  ∃ $\varphi$ "
23 (**Actualist quantification for individuals**)
24  consts Exists::" $\gamma$ " ("existsAt")
25  abbreviation mforallAct::" $\gamma \Rightarrow \sigma$ " ("∀ $\varepsilon$ " [8]9) where "∀ $\varepsilon$  $\Phi$   $\equiv$   $\lambda$ w.  $\forall$ x. (existsAt x w) → ( $\Phi$  x w)"
26  abbreviation mforallActB (binder"∀ $\varepsilon$ "[8]9) where "∀ $\varepsilon$ x.  $\varphi$ (x)  $\equiv$  ∀ $\varepsilon$  $\varphi$ "
27  abbreviation mexistsAct::" $\gamma \Rightarrow \sigma$ " ("∃ $\varepsilon$ " [8]9) where "∃ $\varepsilon$  $\Phi$   $\equiv$   $\lambda$ w.  $\exists$ x. (existsAt x w) ∧ ( $\Phi$  x w)"
28  abbreviation mexistsActB (binder"∃ $\varepsilon$ "[8]9) where "∃ $\varepsilon$ x.  $\varphi$ (x)  $\equiv$  ∃ $\varepsilon$  $\varphi$ "
29 (**Modal operators**)
30  consts aRel::" $\tau$ " (infix "r" 70) (*Accessibility Relation r*)
31  abbreviation mbox::" $\sigma \Rightarrow \sigma$ " ("□" [52]53) where "□ $\varphi$   $\equiv$   $\lambda$ w.  $\forall$ v. (w r v) → ( $\varphi$  v)"
32  abbreviation mdia::" $\sigma \Rightarrow \sigma$ " ("◇" [52]53) where "◇ $\varphi$   $\equiv$   $\lambda$ w.  $\exists$ v. (w r v) ∧ ( $\varphi$  v)"
33 (**Meta-logical predicates**)
34  abbreviation globalvalid::" $\sigma \Rightarrow$ bool" ("⊧" [8]9) where "⊧ $\varphi$   $\equiv$   $\forall$ w. ( $\varphi$  w)"
35 (**Definition of rigidly intensionalised predicates**)
36  ( $\llbracket \varphi \rrbracket$  converts an extensional object  $\varphi$  into 'rigid' intensional one*)
37  abbreviation trivialConversion::"bool  $\Rightarrow$   $\sigma$ " ("⌊" [8]9) where "⌊ $\varphi$   $\equiv$  ( $\lambda$ w.  $\varphi$ )"
38  (* $\llbracket \varphi \rrbracket$ : the extension of a (possibly) non-rigid predicate  $\varphi$  is turned into a rigid intensional one and Q is applied to the latter;  $\llbracket \varphi \rrbracket$  is read as "the rigidly intensionalised predicate  $\varphi$ "*)
39  abbreviation nextPredArg::"( $\gamma \Rightarrow \sigma$ )  $\Rightarrow$   $\gamma \Rightarrow \sigma$ " (infix"⌊"60) where " $\llbracket \varphi \rrbracket$   $\equiv$   $\lambda$ w.  $\varphi$  ( $\lambda$ x. (P x w)) w"
40  lemma "∀ $\varphi$  P.  $\varphi$  P =  $\llbracket \varphi \rrbracket$  P" nitpick[user_axioms] oops (*Countermodel: notions are not the same*)
41  (**Some further definitions required for Fitting's variant**)
42  ( $\varphi$ 's argument is a relativized term of extensional type derived from an intensional predicate*)
43  abbreviation extPredArg::"( $\delta \Rightarrow \sigma$ )  $\Rightarrow$   $\gamma \Rightarrow \sigma$ " (infix"⌊"60) where " $\llbracket \varphi \rrbracket$  P  $\equiv$   $\lambda$ w.  $\varphi$  ( $\lambda$ x. P x w) w"
44  (*Another variant where  $\varphi$  has two arguments (the first one being relativized)*)
45  abbreviation extPredArg1::"( $\delta \Rightarrow \gamma$ )  $\Rightarrow$   $\gamma \Rightarrow \sigma$ " (infix"⌊"60) where " $\llbracket \varphi \rrbracket$ 1 P  $\equiv$   $\lambda$ z.  $\lambda$ w.  $\varphi$  ( $\lambda$ x. P x w) z w"
46  (**Consistency**)
47  lemma True nitpick[satisfy] oops (*Model found by Nitpick*)
48 end

```

Fig. 1. Shallow semantical embedding of IHOML in HOL.

of the freshly introduced symbols; this not only supports a better intuitive understanding of these notions but also reduces the number of polymorphic terms in the formalization (since polymorphism may generally cause decreased proof automation performance).

The world-lifted modal  $\Box$ -operator and the polymorphic, world-lifted universal quantifier  $\forall$ , as already discussed in §2, are introduced in lines 31 and 19, respectively (the 'a' in the type declaration for  $\forall$  represents a type variable). In line 20, user-friendly binder-notation for  $\forall$  is additionally defined. In addition to the (polymorphic) possibilist quantifiers,  $\forall$  and  $\exists$ , defined this way in lines 19–22, further actualist quantifiers,  $\forall^E$  and  $\exists^E$ , are introduced in lines 24–28; their definition is guarded by an explicit, possibly empty, `existsAt` predicate, which encodes whether an individual object actually “exists” at a particular given world, or not. These additional actualist quantifiers are declared non-polymorphic, so that they support quantification over individuals only. In the subsequent analysis of the variants of Gödel’s argument, as contributed by Scott, Anderson and Fitting, we will indeed use  $\forall$  and  $\exists$  for different types in the type hierarchy of HOL, while keeping  $\forall^E$  and  $\exists^E$  for quantification over individuals only.

The notion of global validity of a world-lifted formula  $\psi_\sigma$ , denoted as  $[\psi]$ , is introduced in line 34 as an abbreviation for  $\forall w_1. \psi w$ .

Note that an (intensional) base modal logic **K** is introduced in the theory **IHOML** (Fig. 1). In later sections we will switch to logics **KB** and **S5** by postulating respective conditions (symmetry, and additionally reflexivity and transitivity) on the accessibility relation **r**.

In lines 35–46 some further abbreviations are declared, which address the mediation between intensions and extensions of properties. World-lifted propositions and intensional properties are modeled as terms of types  $\sigma$  and  $\gamma$  respectively, i.e., they are technically handled in HOL as functions over worlds whose extensions are obtained by applying them to a given world  $w$  in context. The operation  $(\llbracket \varphi \rrbracket)$  in line 37 is trivially converting a world-independent proposition of Boolean type into a *rigid* world-lifted proposition of type  $\sigma$ ; the rigid world-lifted propositions obtained from this trivial conversion have identical evaluations in all worlds.

The  $\Downarrow$ -operator in line 40, which is of type  $(\gamma \Rightarrow \sigma) \Rightarrow \gamma \Rightarrow \sigma$ , is slightly more involved. It evaluates its second argument, which is a property  $P$  of type  $\gamma$ , for a given world  $w$ , and it then *rigidly intensionalizes* the obtained extension of  $P$  in  $w$ . For technical reasons, however,  $\Downarrow$  is introduced as a binary operator, with its first argument being a world-lifted predicate  $\varphi_{\gamma \Rightarrow \sigma}$  that is being applied to the rigidly intensionalized  $\Downarrow P_\gamma$ ; in fact, all occurrences of the  $\Downarrow$ -operator in our subsequent sections will have this binary pattern.

The lemma statement in line 41 confirms that intensional properties  $P_\gamma$  are generally different from their rigidly intensionalized counterparts  $\downarrow P_\gamma$ : Isabelle/HOL's model finder Nitpick [14] generates a countermodel to the claim that they are (Leibniz-)equal.

A related (non-bold) binary operator  $\downarrow$ , of type  $(\delta \Rightarrow \sigma) \Rightarrow \gamma \Rightarrow \sigma$ , is introduced in line 44. Its first argument is a predicate  $\varphi_{\delta \Rightarrow \sigma}$  applicable to extensions of properties, and its second argument is an intensional property. The  $\downarrow$ -operator evaluates its second argument  $P_\gamma$  in a given world  $w$ , thereby obtaining an extension  $\downarrow P_\gamma$  of type  $\delta$ , and then it applies its first argument  $\varphi_{\delta \Rightarrow \sigma}$  to this extension. The  $\downarrow_1$ -operator is analogous, but its first argument  $\varphi$  is now of type  $\delta \Rightarrow \gamma$ , which can be understood as world-lifted binary predicate whose first argument is of type  $\delta$  and its second argument of type  $e$ . The  $\downarrow_1$ -operator evaluates the intensional argument  $P_\gamma$ , given to it in second position, in a given world  $w$ , and it then applies  $\varphi_{\delta \Rightarrow (e \Rightarrow \sigma)}$  to the result of this operation and subsequently to its (unmodified) second argument  $z_e$ .

In line 48, consistency of the introduced concepts is confirmed by the model finder Nitpick [14]. Since only abbreviations and no axioms have been introduced so far, the consistency of the Isabelle/HOL theory IHOML in Fig. 3.1 is actually evident.

### 3.2. Filters and Ultrafilters

Two related world-lifted notions of modal filters and modal ultrafilters are defined in Fig. 2; for a general introduction to filters and ultrafilters we refer to the corresponding mathematical literature (e.g. [15]).

$\delta$ -Ultrafilters are introduced in line 26 as world-lifted characteristic functions of type  $(\delta \Rightarrow \sigma) \Rightarrow \sigma$ . They thus denote  $\sigma$ -sets of  $\sigma$ -sets of objects of type  $\delta$ . In other words, a  $\delta$ -Ultrafilter is a  $\sigma$ -subset of the  $\sigma$ -powerset of  $\delta$ -type property extensions.

A  $\delta$ -Ultrafilter  $\phi$  is defined as a  $\delta$ -Filter satisfying an additional maximality condition:  $\forall \varphi. \varphi \in^\delta \phi \vee (-^{1\delta} \varphi) \in^\delta \phi$ , where  $\in^\delta$  is elementhood of  $\delta$ -type objects in  $\sigma$ -sets of  $\delta$ -type objects (see line 4), and where  $-^{1\delta}$  is the relative set complement operation on sets of entities (see line 14).

The notion of  $\delta$ -Filter is introduced in lines 17 and 18. A  $\delta$ -Filter  $\phi$  is required to

- be large:  $\mathbf{U}^\delta \in^\delta \phi$ , where  $\mathbf{U}^\delta$  denotes the full set of  $\delta$ -type objects we start with (see line 8),

```

1 theory ModalUltrafilter imports IHOML
2 begin
3 (**Some abbreviations for operations on  $\delta/\gamma$ -sets in modal context**)
4 abbreviation elem_delta::" $\delta \Rightarrow (\delta \Rightarrow \sigma) \Rightarrow \sigma$ " (infixr" $\in$ "99) where " $x \in^{\delta} S \equiv S x$ "
5 abbreviation elem_gamma::" $\gamma \Rightarrow (\gamma \Rightarrow \sigma) \Rightarrow \sigma$ " (infixr" $\in$ "99) where " $x \in^{\gamma} S \equiv S x$ "
6 abbreviation emptySet_delta::" $\delta$  (" $\emptyset^{\delta}$ ") where " $\emptyset^{\delta} \equiv \lambda x. \text{False}$ "
7 abbreviation emptySet_gamma::" $\gamma$  (" $\emptyset^{\gamma}$ ") where " $\emptyset^{\gamma} \equiv \lambda x. \perp$ "
8 abbreviation fullSet_delta::" $\delta$  (" $U^{\delta}$ ") where " $U^{\delta} \equiv \lambda x. \text{True}$ "
9 abbreviation fullSet_gamma::" $\gamma$  (" $U^{\gamma}$ ") where " $U^{\gamma} \equiv \lambda x. \top$ "
10 abbreviation entails_delta::" $\delta \Rightarrow \delta \Rightarrow \sigma$ " (infixr" $\subseteq$ "51) where " $\varphi \subseteq^{\delta} \psi \equiv \lambda w. \forall x. \varphi x \longrightarrow \psi x$ "
11 abbreviation entails_gamma::" $\gamma \Rightarrow \gamma \Rightarrow \sigma$ " (infixr" $\subseteq$ "51) where " $\varphi \subseteq^{\gamma} \psi \equiv \forall x. \varphi x \longrightarrow \psi x$ "
12 abbreviation andPred_delta::" $\delta \Rightarrow \delta \Rightarrow \delta$ " (infixr" $\cap$ "51) where " $\varphi \cap^{\delta} \psi \equiv \lambda x. \varphi x \wedge \psi x$ "
13 abbreviation andPred_gamma::" $\gamma \Rightarrow \gamma \Rightarrow \gamma$ " (infixr" $\cap$ "51) where " $\varphi \cap^{\gamma} \psi \equiv \lambda x. \varphi x \wedge \psi x$ "
14 abbreviation negpred_delta::" $\delta \Rightarrow \delta$ " (" $\neg^{\delta}$ " [52]53) where " $\neg^{\delta} \equiv \lambda x. \neg(\psi x)$ "
15 abbreviation negpred_gamma::" $\gamma \Rightarrow \gamma$ " (" $\neg^{\gamma}$ " [52]53) where " $\neg^{\gamma} \equiv \lambda x. \neg(\psi x)$ "
16 (**Definition of  $\delta/\gamma$ -Filter in modal context**)
17 abbreviation filter_delta::" $(\delta \Rightarrow \sigma) \Rightarrow \sigma$ " (" $\delta$ -Filter") where " $\delta$ -Filter  $\Phi \equiv$ 
18    $U^{\delta} \in^{\delta} \Phi \wedge \neg \emptyset^{\delta} \in^{\delta} \Phi \wedge (\forall \varphi \psi. (\varphi \in^{\delta} \Phi \wedge \varphi \subseteq^{\delta} \psi) \rightarrow \psi \in^{\delta} \Phi) \wedge (\forall \varphi \psi. (\varphi \in^{\delta} \Phi \wedge \psi \in^{\delta} \Phi) \rightarrow ((\varphi \cap^{\delta} \psi) \in^{\delta} \Phi))$ "
19 abbreviation filter_gamma::" $(\gamma \Rightarrow \sigma) \Rightarrow \sigma$ " (" $\gamma$ -Filter") where " $\gamma$ -Filter  $\Phi \equiv$ 
20    $U^{\gamma} \in^{\gamma} \Phi \wedge \neg \emptyset^{\gamma} \in^{\gamma} \Phi \wedge (\forall \varphi \psi. (\varphi \in^{\gamma} \Phi \wedge \varphi \subseteq^{\gamma} \psi) \rightarrow \psi \in^{\gamma} \Phi) \wedge (\forall \varphi \psi. (\varphi \in^{\gamma} \Phi \wedge \psi \in^{\gamma} \Phi) \rightarrow ((\varphi \cap^{\gamma} \psi) \in^{\gamma} \Phi))$ "
21 (** $\delta/\gamma$ -Filter are consistent**)
22 lemma cons_delta: " $|\forall \Phi \varphi. \delta\text{-Filter } \Phi \rightarrow \neg(\varphi \in^{\delta} \Phi \wedge (\neg^{\delta} \varphi) \in^{\delta} \Phi)|$ " by fastforce
23 lemma cons_gamma: " $|\forall \Phi \varphi. \gamma\text{-Filter } \Phi \rightarrow \neg(\varphi \in^{\gamma} \Phi \wedge (\neg^{\gamma} \varphi) \in^{\gamma} \Phi)|$ " by fastforce
24 (**Definition of  $\delta/\gamma$ -Ultrafilter in modal context**)
25 abbreviation ultrafilter_delta::" $(\delta \Rightarrow \sigma) \Rightarrow \sigma$ " (" $\delta$ -Ultrafilter") where
26   " $\delta$ -Ultrafilter  $\Phi \equiv \delta$ -Filter  $\Phi \wedge (\forall \varphi. \varphi \in^{\delta} \Phi \vee (\neg^{\delta} \varphi) \in^{\delta} \Phi)$ "
27 abbreviation ultrafilter_gamma::" $(\gamma \Rightarrow \sigma) \Rightarrow \sigma$ " (" $\gamma$ -Ultrafilter") where
28   " $\gamma$ -Ultrafilter  $\Phi \equiv \gamma$ -Filter  $\Phi \wedge (\forall \varphi. \varphi \in^{\gamma} \Phi \vee (\neg^{\gamma} \varphi) \in^{\gamma} \Phi)$ "
29 end

```

Fig. 2. Definition of  $\delta/\gamma$ -Filters and  $\delta/\gamma$ -Ultrafilters.

- exclude the empty set:  $\emptyset^{\delta} \notin^{\delta} \phi$ , where  $\emptyset^{\delta}$  is the world-lifted empty set of  $\delta$ -type objects (see line 6),
- be closed under supersets:  $\forall \varphi \psi. (\varphi \in^{\delta} \phi \wedge \varphi \subseteq^{\delta} \psi) \rightarrow \psi \in^{\delta} \phi$  (the world-lifted subset relation  $\subseteq^{\delta}$  is defined in line 10), and
- be closed under intersections:  $\forall \varphi \psi. (\varphi \in^{\delta} \phi \wedge \psi \in^{\delta} \phi) \rightarrow (\varphi \cap^{\delta} \psi) \in^{\delta} \phi$  (the intersection operation  $\cap^{\delta}$  is defined in line 12).

$\gamma$ -Ultrafilters, which are of type  $(\gamma \Rightarrow \sigma) \Rightarrow \sigma$ , are analogously defined as a  $\sigma$ -subset of the  $\sigma$ -powerset of  $\gamma$ -type property extensions.

The distinction of both notions of ultrafilters is needed in our subsequent investigation. This is because we will rigorously distinguish between positive property intensions (as used by Scott and Anderson) and positive property extensions (as utilized by Fitting).

By using polymorphic definitions, several “duplications” of abbreviations in the theory `ModalUltrafilter` (Fig. 2) could be avoided. To support a more precise understanding of  $\delta$ - and  $\gamma$ -Ultrafilters, and their differences, however, we have decided to be very transparent and explicit regarding type information in the provided definitions.

## 4. Scott’s Variant of Gödel’s Argument

Scott’s variant of Gödel’s argument has been reproduced by Fitting in his book [16]. It is Fitting’s formalization of Scott’s variant that we have encoded and verified first in our computer-supported analysis of positive properties, ultrafilters and modal collapse. This encoding of Scott’s variant is presented in Fig. 3 and its presentation is continued in Fig. 4.

Part I of the argument is reconstructed in lines 4-11 of Fig. 3 and verified with automated reasoning tools.<sup>2</sup> In this part we conclude from the premises and definitions (lines 5–8) that a Godlike being possibly exists (theorem T3 in line 11):  $[\diamond\exists^E\mathbf{G}]$ ; this follows from theorems T1 and T2 that are proved in lines 9 and 10. Note that, using binder notation,  $[\diamond\exists^E\mathbf{G}]$  can be more intuitively presented as  $[\diamond\exists^E x.\mathbf{G}x]$ . The most essential definition, the definition of property  $\mathbf{G}$ , which is of type  $\gamma$  and which defines a Godlike being  $x_e$  to possess all (intensional!) positive properties  $\mathcal{P}$ , is given in line 5. Premises that govern the notion of (intensional) positive properties  $\mathcal{P}$  are A1 (which is split into A1a and A1b), A2 and A3; see lines 6–8. Scott [28] actually avoids axiom A3 and instead directly postulates T2 (the sole purpose of A3 is to support T2). Although we here explicitly include the inference from A3 to T2, it could also be left out without any implications for the rest of the proof.

Part II of the argument is presented in lines 12–20. In line 13 we switch from base modal logic  $\mathbf{K}$  to logic  $\mathbf{KB}$  by postulating symmetry of the accessibility relation  $\mathbf{r}$ . Utilizing the same tools as before, and by exploiting theorems T3, T4 and T5, we finally prove, in line 20, the main theorem T6, which states that a Godlike being necessarily exists:  $[\Box\exists^E\mathbf{G}]$ , resp.  $[\Box\exists^E x.\mathbf{G}x]$  using binder notation.

Consistency of the Isabelle/HOL theory `ScottVariant`, as introduced up to here, is confirmed by the model finder Nitpick [14] in line 21 (which constructs a model with one world and one Godlike entity).

---

<sup>2</sup>The automated reasoning tools that are integrated with Isabelle/HOL, and which we utilize in this article, include `metis`, `smt`, `simp`, `blast`, `force`, and `auto`. In fact, in each case where those occur in the presented Isabelle/HOL formalizations, we have actually first used a generic hammer-tool, called `sledgehammer` [13], which calls state-of-the-art ATPs to prove the statements in question fully automatically and without the need for specifying the particularly required premises; `sledgehammer`, in case of success, subsequently attempts to reconstruct the external proofs reported by the ATPs in Isabelle/HOL’s trusted kernel by applying the mentioned automated reasoning tools.

Scott's Axioms and Definitions	
(df.G)	$Gx \equiv \forall Y. \gamma. \mathcal{P} Y \rightarrow Yx$
(A1a)	$\forall X. \mathcal{P}(\neg X) \rightarrow \neg(\mathcal{P} X)$ where $\rightarrow$ is set/predicate negation
(A1b)	$\forall X. \neg(\mathcal{P} X) \rightarrow \mathcal{P}(\neg X)$
(A2)	$\forall XY. (\mathcal{P} X \wedge \Box(\forall^E z. Xz \rightarrow Yz)) \rightarrow \mathcal{P} Y$
(A3)	$\forall ZX. ((\forall Y. ZY \rightarrow \mathcal{P} Y) \wedge \Box(\forall x. Xx \leftrightarrow (\forall Y. ZY \rightarrow Yx))) \rightarrow \mathcal{P} X$
(A4)	$\forall X. \mathcal{P} X \rightarrow \Box(\mathcal{P} X)$
(df.E)	$\mathcal{E} Yx \equiv Yx \wedge (\forall Z. Zx \rightarrow \Box(\forall^E z. Yz \rightarrow Zz))$
(df.NE)	$NE x \equiv \forall Y. \mathcal{E} Yx \rightarrow \Box \exists^E Y$
(A5)	$\mathcal{P} NE$

```

1 theory ScottVariant imports IHOML ModalUltrafilter
2 begin
3 (**Positiveness**) consts positiveProperty: "γ⇒σ" ("P")
4 (**Part I**)
5 (*D1*) definition G::γ ("G") where "G x ≡ ∀Y. P Y → Y x"
6 (*A1*) axiomatization where A1a: "[∀X. P(¬X) → ¬(P X)]" and A1b: "[∀X. ¬(P X) → P(¬X)]" and
7 (*A2*) A2: "[∀X Y. (P X ∧ □(∀Ez. X z → Y z)) → P Y]" and
8 (*A3*) A3: "[∀Z X. ((∀Y. Z Y → P Y) ∧ □(∀x. X x ↔ (∀Y. Z Y → Y x))) → P X]"
9 (*T1*) theorem T1: "[∀X. P X → □∃E X]" by (metis A1a A2)
10 (*T2*) theorem T2: "[P G]" by (smt A3 G def)
11 (*T3*) theorem T3: "[□∃E G]" using T1 T2 by simp
12 (**Part II**)
13 (*Logic KB*) axiomatization where symm: "∀x y. x r y → y r x"
14 (*A4*) axiomatization where A4: "[∀X. P X → □(P X)]"
15 (*D2*) definition ess::"γ⇒γ" ("E") where "E Y x ≡ Y x ∧ (∀Z. Z x → □(∀Ez. Y z → Z z))"
16 (*T4*) theorem T4: "[∀X. G x → (E G x)]" by (metis A1b A4 G def ess_def)
17 (*D3*) definition NE::"γ ("NE") where "NE x ≡ ∀Y. E Y x → □∃E Y"
18 (*A5*) axiomatization where A5: "[P NE]"
19 (*T5*) theorem T5: "[□∃E G → □∃E G]" by (metis A5 G_def NE_def T4 symm)
20 (*T6*) theorem T6: "[□∃E G]" using T3 T5 by blast
21 (**Consistency**) lemma True nitpick[satisfy,user_axioms] oops (*Model found by Nitpick*)
22 (**Modal collapse**)
23 lemma MC: "[∀Φ. (Φ → □(Φ))]"
24 proof - {fix w fix Q
25   have "∀x. G x w → (∀Z. Z x → □(∀Ez. G z → Z z)) w" by (metis A1b A4 G_def)
26   hence 1: "(∃x. G x w) → ((Q → □(∀Ez. G z → Q)) w)" by force
27   have "∃x. G x w" using T3 T6 symm by blast
28   hence "(Q → □Q) w" using 1 T6 by blast}
29   thus ?thesis by auto qed
30 (**Analysis of positive properties using ultrafilters**)

```

Fig. 3. Scott's variant of Gödel's argument, following Fitting [16].

In lines 23–29 modal collapse is proved. This is one of the rare cases in our experiments where direct proof automation with Isabelle/HOL's integrated automated reasoning tools (incl. sledgehammer [13]) still fails. A little interactive help is needed here to show that modal collapse indeed follows from the premises in Scott's variant of Gödel's argument.

```

30 (**Analysis of positive properties using ultrafilters**)
31 (*unimportant*) declare [[smt_solver = cvc4, smt_oracle]]
32 (*U1*) theorem U1: "[ $\gamma$ -Ultrafilter  $\mathcal{P}$ ]" (*Sledgehammer succeeds, reconstruction fails*)
33   proof - have 1: "[ $U \in \mathcal{P} \wedge \neg \emptyset \in \mathcal{P}$ ]" using A1b T1 by auto
34   have 2: "[ $\forall \psi \psi. (\varphi \in \mathcal{P} \wedge \varphi \subseteq \psi) \rightarrow \psi \in \mathcal{P}$ ]" by (metis A1b G_def T1 T6 symm)
35   have 3: "[ $\forall \varphi \psi. (\varphi \in \mathcal{P} \wedge \psi \in \mathcal{P}) \rightarrow ((\varphi \cap \psi) \in \mathcal{P})$ ]" by (metis A1b G_def T1 T6 symm)
36   have 4: "[ $\forall \varphi. \varphi \in \mathcal{P} \vee (\neg \varphi) \in \mathcal{P}$ ]" using A1b by blast
37   thus ?thesis by (simp add: 1 2 3 4) qed
38 (*U2*) abbreviation " $\mathcal{P}' \equiv \lambda \varphi. (\mathcal{P} \downarrow \varphi)$ " (*Set of  $\varphi$ 's whose rigidly intens. extensions are positive*)
39 theorem U2: "[ $\gamma$ -Ultrafilter  $\mathcal{P}'$ ]" (*Sledgehammer succeeds, reconstruction fails*)
40   proof - have 1: "[ $U \in \mathcal{P}' \wedge \neg \emptyset \in \mathcal{P}'$ ]" using A1b T1 by auto
41   have 2: "[ $\forall \varphi \psi. (\varphi \in \mathcal{P}' \wedge \varphi \subseteq \psi) \rightarrow \psi \in \mathcal{P}'$ ]" by (metis A1b G_def T1 T6 symm)
42   have 3: "[ $\forall \varphi \psi. (\varphi \in \mathcal{P}' \wedge \psi \in \mathcal{P}') \rightarrow ((\varphi \cap \psi) \in \mathcal{P}')$ ]" by (metis A1b G_def T1 T6 symm)
43   have 4: "[ $\forall \varphi. \varphi \in \mathcal{P}' \vee (\neg \varphi) \in \mathcal{P}'$ ]" using A1b by blast
44   thus ?thesis by (simp add: 1 2 3 4) qed
45 (*U3*) theorem U3: "[ $\forall \varphi. \varphi \in \mathcal{P}' \leftrightarrow \varphi \in \mathcal{P}$ ]" by (metis A1b G_def T1 T6 symm) (* $\mathcal{P}'$  and  $\mathcal{P}$  are equal*)
46 (**Modal Logic S5**)
47 axiomatization where refl: " $\forall x. x \ r \ x$ " and trans: " $\forall x \ y \ z. x \ r \ y \ \wedge \ y \ r \ z \rightarrow x \ r \ z$ "
48 lemma True nitpick[satisfy,user_axioms] oops (*Model found by Nitpick*)
49 (**Barcan and converse Barcan formula for individuals (type e) and properties (type e $\Rightarrow$ i $\Rightarrow$ bool)**)
50 lemma BBind1: "[ $(\forall^E x::e. \Box(\varphi \ x)) \rightarrow \Box(\forall^E x::e. \varphi \ x)$ ]" using MC symm by blast
51 lemma CBind1: "[ $\Box(\forall^E x::e. \varphi \ x) \rightarrow (\forall^E x::e. \Box(\varphi \ x))$ ]" using MC by blast
52 lemma Bpred1: "[ $(\forall x::e \Rightarrow i \Rightarrow \text{bool}. \Box(\varphi \ x)) \rightarrow \Box(\forall x::e \Rightarrow i \Rightarrow \text{bool}. \varphi \ x)$ ]" by simp
53 lemma CBpred1: "[ $\Box(\forall x::e \Rightarrow i \Rightarrow \text{bool}. \varphi \ x) \rightarrow (\forall x::e \Rightarrow i \Rightarrow \text{bool}. \Box(\varphi \ x))$ ]" by simp
54 end

```

Fig. 4. Ultrafilter-analysis of Scott’s variant (continued from Fig. 3).

For more background information and details on the formalization of Scott’s argument, and also on the arguments by Anderson and Fitting as presented in the following sections, we refer to Fitting’s book [16, §11] and our previous work [17].

#### 4.1. Positive Properties and Ultrafilters: Scott

Interesting findings regarding positive properties and ultrafilters in Scott’s variant are revealed in Fig. 4.

Theorem U1, which is proved in lines 32–37, states that the set of positive properties  $\mathcal{P}$  in Scott’s variant constitutes a  $\gamma$ -Ultrafilter.

In line 38, a modified notion of positive properties  $\mathcal{P}'$  is defined as the set of properties  $\varphi$  whose rigidly intensionalized extensions  $\downarrow\varphi$  are in  $\mathcal{P}$ . It is then shown in theorem U2 (lines 39–44), that also  $\mathcal{P}'$  constitutes a  $\gamma$ -Ultrafilter. And theorem U3 in line 45 shows that these two sets,  $\mathcal{P}$  and  $\mathcal{P}'$ , are in fact equal.

In line 47 we switch from logic **KB** to logic **S5** by postulating reflexivity and transitivity of the accessibility relation  $r$  in addition to symmetry (line 13 in Fig. 3); and we show consistency again (line 48). In the re-

maining lines 49–53 in Fig. 4 we show that the Barcan and the converse Barcan formulas are valid for types  $\mathbf{e}$  and  $\gamma$ ; we use for the former type actualist quantifiers (as in the argument) and for the latter type possibilist quantifiers.

## 5. Anderson’s Variant of Gödel’s Argument

Anderson’s variant of Gödel’s argument is presented in Fig. 5.

A central change in comparison to Scott’s variant concerns Scott’s premisses A1a and A1b. Anderson drops A1b and only keeps A1a: “If a property is positive, then its negation is not positive”. This modification, however, has the effect that the necessary existence of a Godlike being would no longer follow (and the reasoning tools in Isabelle/HOL can confirm this; not shown here). Anderson’s variant therefore introduces further emendations: it strengthens the notions of Godlikeness (in line 5) and essence (in line 14). The emended notions, referred to by  $\mathbf{G}^A$  and  $\mathcal{E}^A$ , are as follows:

$\mathbf{G}^A$  An individual  $x$  is Godlike  $\mathbf{G}^A$  if and only if all and only the necessary/essential properties of  $x$  are positive, i.e.,  $\mathbf{G}^A x \equiv \forall Y (\mathcal{P}Y \leftrightarrow \Box(Yx))$ .

$\mathcal{E}^A$  A property  $Y$  is an essence  $\mathcal{E}^A$  of an individual  $x$  if and only if all of  $x$ ’s necessary/essential properties are entailed by  $Y$  and (conversely) all properties entailed by  $Y$  are necessary/essential properties of  $x$ .

As is shown in lines 3–19, no further modifications are required to ensure that the intended theorem T6, the necessary existence of a  $\mathbf{G}^A$ -like being, can (again) be proved.<sup>3</sup>

In line 20, the model finder Nitpick confirms that modal collapse is indeed countersatisfiable in Anderson’s variant of Gödel’s argument. As expected, the reported countermodel consists of two worlds and one entity.

Consistency of theory **AndersonVariant** is confirmed by Nitpick in line 21, by finding a model with only one world and one entity (not shown).

---

<sup>3</sup>In a very stringent interpretation this statement is not entirely true: Theorem T2 in Scott’s argument, which was derived in Fig. 3 from axiom A3 and the definition of  $\mathbf{G}$ , is now directly postulated here (for simplicity reasons) and axiom A3, which had no other purpose besides supporting T2, is dropped. This simplification, however, is obviously independent from the aspects as discussed.

## Anderson's Axioms and Definitions

(df. $G^A$ )	$G^A x \equiv \forall Y. \gamma. \mathcal{P} Y \leftrightarrow \Box(Yx)$
(A1a)	$\forall X. \mathcal{P}(\rightarrow X) \rightarrow \neg(\mathcal{P} X)$ where $\rightarrow$ is set/predicate negation
(A2)	$\forall X Y. (\mathcal{P} X \wedge \Box(\forall^E z. Xz \rightarrow Yz)) \rightarrow \mathcal{P} Y$
(T2)	$\mathcal{P} G^A$
(A4)	$\forall X. \mathcal{P} X \rightarrow \Box(\mathcal{P} X)$
(df. $\mathcal{E}^A$ )	$\mathcal{E}^A Y x \equiv \forall Z. \Box(Zx) \leftrightarrow \Box(\forall^E z. Yz \rightarrow Zz)$
(df. $NE^A$ )	$NE^A x \equiv \forall Y. \mathcal{E}^A Y x \rightarrow \Box \exists^E Y$
(A5)	$\mathcal{P} NE^A$

```

1 theory AndersonVariant imports IHOML ModalUltrafilter
2 begin
3 (**Positiveness**) consts positiveProperty: "γ⇒σ" ("P")
4 (**Part I**)
5 (*D1*) definition GA: γ ("GA") where "GA x ≡ ∀Y: γ. P Y ↔ □(Y x)"
6 (*A1a*) axiomatization where Ala: "[∀X. P(→X) → ¬(P X)]"
7 (*A2*) axiomatization where A2: "[∀X Y. (P X ∧ □(∀Ez. Xz → Yz)) → P Y]"
8 (*T1*) theorem T1: "[∀X. P X → □∃EX]" using Ala A2 by metis
9 (*T2*) axiomatization where T2: "[P GA]" (*there we postulate T2 instead of proving it*)
10 (*T3*) theorem T3: "[□∃E GA]" by (metis Ala A2 T2)
11 (**Part II**)
12 (*Logic KB*) axiomatization where symm: "∀x y. x r y → y r x"
13 (*A4*) axiomatization where A4: "[∀X. P X → □(P X)]"
14 (*D2*) abbreviation essA: "γ⇒γ" ("EA") where "EA Y x ≡ ∀Z. □(Z x) ↔ □(∀Ez. Yz → Zz)"
15 (*T4*) theorem T4: "[∀x. GA x → (EA GA x)]" by (metis (mono_tags) A2 GA def T2 symm)
16 (*D3*) abbreviation NEA: "γ ("NEA") where "NEA x ≡ ∀Y. EA Y x → □∃E Y"
17 (*A5*) axiomatization where A5: "[P NEA]"
18 (*T5*) theorem T5: "[□∃E GA] → [□∃E NEA]" by (metis A2 GA def T2 symm)
19 (*T6*) theorem T6: "[□∃E GA]" using T3 T5 by blast
20 (**Modal collapse**) lemma "[∀φ. (φ → □φ)]" nitpick[user_axioms, show_all] oops (*Countermodel*)
21 (**Consistency**) lemma True nitpick[satisfy, user_axioms] oops (*Model found by Nitpick*)
22 (**Analysis of positive properties using ultrafilters**)
23 (*U1*) theorem U1: "[γ-Ultrafilter P]" nitpick[user_axioms, show_all] oops (*Countermodel*)
24 (*U2*) abbreviation "P' ≡ λφ. (P ↓ φ)" (*Set of φ's whose rigidly intns. extensions are positive*)
25 theorem U2: "[γ-Ultrafilter P']" (*Sledgehammer succeeds, reconstruction fails*)
26 proof - have 1: "[U ∈ P' ∧ ¬∅ ∈ P']" using A2 T1 T2 by fastforce
27 have 2: "[∀φ ψ. (φ ∈ P' ∧ φ ⊆ ψ) → ψ ∈ P']" by (metis A2)
28 have 3: "[∀φ ψ. (φ ∈ P' ∧ ψ ∈ P') → ((φ ∩ ψ) ∈ P')]" by (smt GA def T3 T5 symm)
29 have 4: "[∀φ. φ ∈ P' ∨ (¬φ) ∈ P']" by (smt GA def T3 T5 symm)
30 thus ?thesis by (simp add: 1 2 3 4) qed
31 (*U3*) theorem U3: "[∀φ. φ ∈ P' ↔ φ ∈ P]" nitpick[user_axioms] oops (*Counterterm.: P, P not equal*)
32 (**Modal logic SS: consistency and modal collapse**)
33 axiomatization where refl: "∀x. x r x" and trans: "∀x y z. x r y ∧ y r z → x r z"
34 lemma True nitpick[satisfy, user_axioms] oops (*Model found by Nitpick*)
35 lemma MC: "[∀φ. (φ → □φ)]" nitpick[user_axioms] oops (*Countermodel*)
36 (**Barcan and converse Barcan formula for individuals (type e) and properties (type e⇒i⇒bool**))
37 lemma Bind1: "[(∀Ex::e. □(φ x)) → □(∀Ex::e. φ x)]" nitpick[user_axioms] oops (*Countermodel*)
38 lemma CBind1: "[□(∀Ex::e. φ x) → (∀Ex::e. □(φ x))]" nitpick[user_axioms] oops (*Countermodel*)
39 lemma Bpred1: "[(∀x::e⇒i⇒bool. □(φ x)) → □(∀x::e⇒i⇒bool. φ x)]" by simp
40 lemma CBpred1: "[□(∀x::e⇒i⇒bool. φ x) → (∀x::e⇒i⇒bool. □(φ x))]" by simp
41 end

```

Fig. 5. Anderson's variant of Gödel's argument, following Fitting [16].

### 5.1. Positive Properties and Ultrafilters: Anderson

Regarding positive properties and ultrafilters an interesting difference to our prior observations for Scott's version is revealed by the automated reasoning tools: the set of positive properties  $\mathcal{P}$  in Anderson's variant does *not* constitute a  $\gamma$ -Ultrafilter; Nitpick finds a countermodel to statement U1 in line 23 that consists of two worlds and one entity. However, the modified notion  $\mathcal{P}'$ , i.e., the set of all properties  $\varphi$ , whose rigidly intensionalized extensions are in  $\mathcal{P}$  (line 24), still is a  $\gamma$ -Ultrafilter; see theorem U2, which is proved in lines 25–30. Consequently, the sets  $\mathcal{P}$  and  $\mathcal{P}'$  are not generally equal anymore and Nitpick reports a countermodel for statement U3 in line 31.

In lines 32–40, we once again switch from logic **KB** to logic **S5**, we again show consistency, and we again analyze the Barcan and the converse Barcan formulas for types **e** and  $\gamma$ . In contrast to before, the Barcan and converse Barcan formulas for type **e**, when formulated with actualist quantifiers, are not valid anymore; Nitpick presents countermodels with two worlds and two entities.

## 6. Fitting's Variant of Gödel's Argument

In Fitting's variant of Gödel's Argument, see Fig. 6, the notion of positive properties  $\mathcal{P}$  in the definition of Godlikeness **G** ranges over extensions of properties, i.e., over terms of type  $\delta$ , and not over  $\gamma$ -type intensional properties as in Scott's and Anderson's variants. In Fitting's understanding, positive properties are thus fixed from world to world, while they are world-dependent in Scott's and Anderson's. In technical terms, Scott (resp. Gödel) defines **G** $x$  as  $\forall Y_\gamma. \mathcal{P}Y \rightarrow Yx$  (line 5 in Fig. 3), whereas Fitting modifies this into  $\forall Y_\delta. \mathcal{P}Y \rightarrow \langle Yx \rangle$  (line 5 in Fig. 6). In an analogous way, the notion of essence is emended by Fitting: in Scott's variant, see line 15 in Fig. 3,  $\mathcal{E}Yx$  is defined as  $Yx \wedge (\forall Z. Zx \rightarrow \Box(\forall^E z. Yz \rightarrow Zz))$ , while it becomes  $\langle Yx \rangle \wedge (\forall Z. \langle Zx \rangle \rightarrow \Box(\forall^E z. \langle Yz \rangle \rightarrow \langle Zz \rangle))$  in Fitting's variant (see line 15 in Fig. 6).

The definition of necessary existence NE in line 17 is adapted accordingly, and in several other places of Fitting's variant respective emendations are required to suitably address his alternative interpretation of Gödel's notion of positive properties (see, e.g., theorem T2 in line 9 or axiom A5 in

### Fitting's Axioms and Definitions

(df.G)	$Gx \equiv \forall Y \delta. \mathcal{P} Y \rightarrow (Yx)$	
(A1a)	$\forall X. \mathcal{P}(\neg X) \rightarrow \neg(\mathcal{P} X)$	where $\rightarrow$ is set/predicate negation
(A1b)	$\forall X. \neg(\mathcal{P} X) \rightarrow \mathcal{P}(\neg X)$	
(A2)	$\forall XY. (\mathcal{P} X \wedge \Box(\forall^E z. (Xz) \rightarrow (Yz))) \rightarrow \mathcal{P} Y$	
(T2)	$\mathcal{P} \downarrow G$	
(df.E)	$\mathcal{E} Y x \equiv (Yx) \wedge (\forall Z. (Zx) \rightarrow \Box(\forall^E z. (Yz) \rightarrow (Zz)))$	
(df.NE)	$NE x \equiv \forall Y. \mathcal{E} Y x \rightarrow \Box(\exists^E z. (Yz))$	
(A5)	$\mathcal{P} \downarrow NE$	

```

1 theory FittingVariant imports IHOML ModalUltrafilter
2 begin
3 (**Positiveness**) consts Positiveness: "δ⇒σ" ("P")
4 (**Part I**)
5 (*D1*) abbreviation God: γ ("G") where "G x ≡ ∀Y:δ. P Y → (Y x)"
6 (*A1*) axiomatization where A1a: "[∀X. P(¬X) → ¬(P X)]" and A1b: "[∀X. ¬(P X) → P(¬X)]"
7 (*A2*) axiomatization where A2: "[∀X Y. (P X ∧ (□(∀Ez. (X z) → (Y z)))) → P Y]"
8 (*T1*) theorem T1: "[∀X. P X → □(∃Ez. (X z))]" using A1a A2 by blast
9 (*T2*) axiomatization where T2: "[P↓G]"
10 (*T3*) theorem T3deRe: "[(λX. □∃E X) ↓G]" using T1 T2 by simp
11 theorem T3deDicto: "[□∃E ↓G]" nitpick[user_axioms] oops (**Countermodel*)
12 (**Part II*)
13 (*Logic KB*) axiomatization where symm: "∀x y. x r y → y r x"
14 (*A4*) axiomatization where A4: "[∀X. P X → □(P X)]"
15 (*D2*) abbreviation ess: "δ⇒γ" ("E") where "E Y x ≡ (Y x) ∧ (∀Z. (Z x) → (□(∀Ez. (Y z) → (Z z))))"
16 (*T4*) theorem T4: "[∀X. G x → ((E ↓G) x)]" using A1b by metis
17 (*D3*) definition NE: γ ("NE") where "NE x ≡ ∀Y. E Y x → □(∃Ez. (Y z))"
18 (*A5*) axiomatization where A5: "[P↓NE]"
19 lemma help1: "[∃ ↓G → □∃E ↓G]" sorry (**...longer interactive proof, omitted here...*)
20 lemma help2: "[∃ ↓G → ((λX. □∃E X) ↓G)]" by (metis A4 help1)
21 (*T5*) theorem T5deDicto: "[□∃E ↓G] → [□∃E ↓G]" using T3deRe help1 by blast
22 theorem T5deRe: "[((λX. □∃E X) ↓G) → ((λX. □∃E X) ↓G)]" by (metis help2)
23 (*T6*) theorem T6deDicto: "[□∃E ↓G]" using T3deRe help1 by blast
24 theorem T6deRe: "[((λX. □∃E X) ↓G) → ((λX. □∃E X) ↓G)]" using T3deRe help2 by blast
25 (**Modal collapse*) lemma MC: "[∀Φ. (Φ → (□ Φ))]" nitpick[user_axioms] oops (**Countermodel*)
26 (**Consistency*) lemma True nitpick[satisfy,user_axioms] oops (**Model found by Nitpick*)
27 (**Analysis of positive properties using ultrafilters*)
28 (*U1*) theorem U1: "[δ-Ultrafilter P]" (*Sledgehammer succeeds, reconstruction fails*)
29 proof - have 1: "[U ∈P P ∧ ¬θ ∈P P]" using A1b T1 by auto
30 have 2: "[∀φ ψ. (φ ∈P P ∧ φ ⊆P ψ) → ψ ∈P P]" by (metis A1b T3deRe)
31 have 3: "[∀φ ψ. (φ ∈P P ∧ ψ ∈P P) → ((φ ∩P ψ) ∈P P)]" by (metis A1b T3deRe)
32 have 4: "[∀φ. φ ∈P P ∨ (¬φ) ∈P P]" using A1b by blast
33 thus ?thesis by (simp add: 1 2 3 4) qed
34 (**Modal logic S5: consistency and modal collapse*)
35 axiomatization where refl: "∀x. x r x" and trans: "∀x y z. x r y ∧ y r z → x r z"
36 lemma True nitpick[satisfy,user_axioms] oops (**Model found by Nitpick*)
37 lemma MC: "[∀Φ. (Φ → (□ Φ))]" nitpick[user_axioms] oops (**Countermodel*)
38 (**Barcan and converse Barcan formula for individuals (type e) and properties (type e⇒bool**)
39 lemma B1nd1: "[(∀Ex: e. □(φ x)) → □(∀Ex: e. φ x)]" nitpick[user_axioms] oops (**Countermodel*)
40 lemma B1nd2: "[□(∀Ex: e. φ x) → (∀Ex: e. □(φ x))]" nitpick[user_axioms] oops (**Countermodel*)
41 lemma Bpred1: "[(∀x: e⇒i⇒bool. □(φ x)) → □(∀x: e⇒i⇒bool. φ x)]" by simp
42 lemma Bpred2: "[□(∀x: e⇒i⇒bool. φ x) → (∀x: e⇒i⇒bool. □(φ x))]" by simp
43 end

```

Fig. 6. Fitting's variant of Gödel's argument.

line 18). Fitting’s expressive logical system (IHOML) also allows us to distinguish between *de dicto* and *de re* readings of theorems T3, T5, and T6. Except for the *de dicto* reading of T3, which has a countermodel with two worlds and two entities, all of these statements are proved automatically by the reasoning tools integrated with Isabelle/HOL.

As intended by Fitting, modal collapse is not provable anymore, which can be seen in line 25, where Nitpick reports a countermodel with two worlds and one entity.

Consistency of the Isabelle/HOL theory `FittingVariant`, as introduced up to here, is confirmed by Nitpick in line 26 (one world, one entity).

### 6.1. Positive Properties and Ultrafilters: Fitting

The type of  $\mathcal{P}$  has changed in Fitting’s variant from the prior  $\gamma \Rightarrow \sigma$  to  $\delta \Rightarrow \sigma$ . Hence, in our ultrafilter analysis, the notion of a  $\gamma$ -Ultrafilter no longer applies and we must consult the corresponding notion of a  $\delta$ -Ultrafilter. Theorem U1, which is proved in lines 28–33 of Fig. 6, confirms that Fitting’s emended notion of  $\mathcal{P}$  indeed constitutes a  $\delta$ -Ultrafilter.

In line 35 we again switch from modal logic **KB** to logic **S5**. Consistency of the Isabelle/HOL theory `FittingVariant` in S5 is confirmed in line 36, and countersatisfiability of modal collapse is reconfirmed in line 37.

Moreover, like for Anderson’s variant before, we get a countermodel for the Barcan formula and the converse Barcan formula on type **e**, when formulated with actualist quantifiers. The Barcan formula and its converse are proved valid for type  $\gamma$ .

## 7. Conclusion

Anderson and Fitting both succeed in altering Gödel’s modal ontological argument in such a way that the intended result, the necessary existence of a Godlike being, is maintained while modal collapse is avoided. And both solutions, from a cursory reading, are quite different.

We conclude by rephrasing in more precise, technical terms what has been mentioned at abstract level already in the mentioned related article [24, §2.3]:

In order to compare the argument variants by Scott, Anderson, and Fitting, two notions of ultrafilters were formalized in Isabelle/HOL: A  $\delta$ -Ultrafilter, of type  $(\delta \Rightarrow \sigma) \Rightarrow \sigma$ , is defined on the powerset of individuals,

i.e., on the set of rigid properties, and a  $\gamma$ -Ultrafilter, which is of type  $(\delta \Rightarrow \sigma) \Rightarrow \sigma$ , is defined on the powerset of concepts, i.e., on the set of non-rigid, world-dependent properties. In our formalizations of the variants, a careful distinction was made between the original notion of a positive property  $\mathcal{P}$  that applies to (intensional) properties and a restricted notion  $\mathcal{P}'$  that applies to properties whose rigidified extensions are  $\mathcal{P}$ -positive. Using these definitions the following results were proved computationally:

- In Scott's variant both  $\mathcal{P}$  and  $\mathcal{P}'$  coincide, and both are  $\gamma$ -Ultrafilters.
- In Anderson's variant  $\mathcal{P}$  and  $\mathcal{P}'$  do not coincide, and only  $\mathcal{P}'$ , but not  $\mathcal{P}$ , is a  $\gamma$ -Ultrafilter.
- In Fitting's variant, the  $\mathcal{P}$  in the sense of Scott and Anderson is not considered an appropriate notion. However, Fitting's emended notion of a positive property  $\mathcal{P}$ , which applies to extensions of properties, corresponds to our definition of  $\mathcal{P}'$  in Scott's and Anderson's variants; and, as was to be expected, Fitting's emended notion of  $\mathcal{P}$  constitutes a  $\delta$ -Ultrafilter.

The presented computational experiments thus reveal an intriguing correspondence between the variants of the ontological argument by Anderson and Fitting, which otherwise seem quite different. The variants of Anderson and Fitting require that only the restricted notion of a positive property is an ultrafilter.

The notion of positive properties in Gödel's ontological argument is thus aligned with the mathematical notion of a (principal) *modal* ultrafilter on intensional properties, and to avoid modal collapse it is sufficient to restrict the modal ultrafilter-criterion to property extensions. In a sense, the notion of Godlike being “ $G_x$ ” of Gödel is thus in close correspondence to the  $x$ -object in a principal modal ultrafilter “ $F_x$ ” of positive properties. This appears interesting and relevant, since metaphysical existence of a Godlike being is now linked to existence of an abstract object in a mathematical theory.

Further research could look into a formal analysis of monotheism and polytheism for the studied variants of Gödel's ontological argument. We conjecture that different notions of equality will eventually support both views, and a respective formal exploration study could take Kordula Świątorzecka's related work [31] as a starting point.

**Acknowledgements.** This work was supported by VolkswagenStiftung under grant CRAP (Consistent Rational Argumentation in Politics). As already mentioned, the technical results presented in this article have been summarized at abstract level in a joint article with Daniel Kirchner and Ed Zalta. We are also grateful to the anonymous reviewer.

## References

- [1] C. A. Anderson, *Some emendations of Gödel's ontological proof*, **Faith and Philosophy**, vol. 7(3) (1990), pp. 291–303.
- [2] C. A. Anderson, M. Gettings, *Gödel's ontological proof revisited*, [in:] **Gödel'96: Logical Foundations of Mathematics, Computer Science, and Physics: Lecture Notes in Logic 6**, pp. 167–172, Springer (1996).
- [3] C. Benzmüller, *Computational metaphysics: New insights on Gödel's ontological argument and modal collapse*, [in:] In S. Kovač, K. Świątorzecka (eds.), **Formal Methods and Science in Philosophy III, Informal Proceedings**, Dubrovnik, Croatia (2019), pp. 3–4.
- [4] C. Benzmüller, *Universal (meta-)logical reasoning: Recent successes*, **Science of Computer Programming**, vol. 172 (2019), pp. 48–62.
- [5] C. Benzmüller, P. Andrews, *Church's type theory*, [in:] E. N. Zalta (ed.), **The Stanford Encyclopedia of Philosophy**, Metaphysics Research Lab, Stanford University (2019), pp. 1–62 (in pdf version).
- [6] C. Benzmüller, D. Fuenmayor, *Can computers help to sharpen our understanding of ontological arguments?*, [in:] S. Gosh, R. Uppalari, K. V. Rao, V. Agarwal, S. Sharma (eds.), **Mathematics and Reality, Proceedings of the 11th All India Students' Conference on Science & Spiritual Quest (AISSQ), 6–7 October, 2018, IIT Bhubaneswar, Bhubaneswar, India**, The Bhaktivedanta Institute, Kolkata (2018), pp. 95–226.
- [7] C. Benzmüller, L. Paulson, *Quantified multimodal logics in simple type theory*, **Logica Universalis**, vol. 7(1) (2013), pp. 7–20.
- [8] C. Benzmüller, N. Sultana, L. C. Paulson, F. Theiß, *The higher-order prover LEO-II*, **Journal of Automated Reasoning**, vol. 55(4) (2015), pp. 389–404.

- [9] C. Benz Müller, B. Woltzenlogel Paleo, *Automating Gödel's ontological proof of God's existence with higher-order automated theorem provers*, [in:] T. Schaub, G. Friedrich, B. O'Sullivan (eds.), **ECAI 2014, Frontiers in Artificial Intelligence and Applications**, vol. 263 (2014), pp. 93–98.
- [10] C. Benz Müller, B. Woltzenlogel Paleo, *Interacting with modal logics in the Coq proof assistant*, [in:] L. D. Beklemishev, D. V. Musatov (eds.), **Computer Science – Theory and Applications – 10th International Computer Science Symposium in Russia, CSR 2015, Listvyanka, Russia, July 13–17, 2015, Proceedings, Lecture Notes in Computer Science**, vol. 9139 (2015), pp. 398–411.
- [11] C. Benz Müller, B. Woltzenlogel Paleo, *The inconsistency in Gödel's ontological argument: A success story for AI in metaphysics*, [in:] S. Kambhampati (ed.), **IJCAI 2016**, vol. 1–3, AAAI Press (2016), pp. 936–942.
- [12] C. Benz Müller, B. Woltzenlogel Paleo, *An object-logic explanation for the inconsistency in Gödel's ontological theory* (extended abstract, sister conferences), [in:] M. Helmert, F. Wotawa (eds.), **KI 2016: Advances in Artificial Intelligence, Proceedings**, Lecture Notes in Computer Science, vol. 6172, Springer, Berlin (2016), pp. 244–250.
- [13] J. C. Blanchette, S. Böhme, L. C. Paulson, *Extending Sledgehammer with SMT solvers*, **Journal of Automated Reasoning**, vol. 51(1) (2013), pp. 109–128.
- [14] J. C. Blanchette, Tobias Nipkow, *Nitpick: A counterexample generator for higher-order logic based on a relational model finder*, [in:] M. Kaufmann, L. C. Paulson (eds.) **Interactive Theorem Proving, First International Conference, ITP 2010, Edinburgh, UK, July 11–14, 2010. Proceedings**, Lecture Notes in Computer Science, vol. 6172, Springer, Berlin (2010), pp. 131–146.
- [15] S. Burris, H. P. Sankappanavar, **A course in universal algebra**, Springer, New York, Heidelberg, Berlin (2012).
- [16] M. Fitting, **Types, Tableaus, and Gödel's God**, Kluwer, Dordrecht (2002).
- [17] D. Fuenmayor, C. Benz Müller, *Automating emendations of the ontological argument in intensional higher-order modal logic*, [in:] **KI 2017: Advances in Artificial Intelligence 40th Annual German Conference on AI, Dortmund, Germany, September 25–29, 2017, Proceedings**, Lecture Notes in Artificial Intelligence, vol. 10505, Springer (2017) pp. 114–127.

- [18] D. Fuenmayor, C. Benz Müller, *Types, Tableaus and Gödel's God in Isabelle/HOL*, **Archive of Formal Proofs** (2017), pp. 1–34. Note: verified data publication.
- [19] J. Garson, *Modal logic*, [in:] E. N. Zalta (ed.), **The Stanford Encyclopedia of Philosophy**, Metaphysics Research Lab, Stanford University (2018).
- [20] K. Gödel, *Appendix A. Notes in Kurt Gödel's Hand*, [in:] J. H. Sobel (ed.), **Logic and Theism: Arguments for and Against Beliefs in God**, Cambridge University Press, Cambridge (1970), pp. 144–145.
- [21] P. Hájek, *Magari and others on Gödel's ontological proof*, [in:] A. Ursini, P. Agliano (eds.), **Logic and algebra**, Dekker, New York (1996), pp. 125–135.
- [22] P. Hájek, *Der Mathematiker und die Frage der Existenz Gottes*, [in:] B. Buldt et al. (eds.), **Kurt Gödel. Wahrheit und Beweisbarkeit**, öbv & hpt Verlagsgesellschaft mbH, Wien (2001), pp. 325–336.
- [23] P. Hájek, *A new small emendation of Gödel's ontological proof*, **Studia Logica**, vol. 71(2) (2002), pp. 149–164.
- [24] D. Kirchner, C. Benz Müller, E. N. Zalta, *Computer science and metaphysics: A cross-fertilization*, **Open Philosophy**, vol. 2 (2019), pp. 230–251.
- [25] S. Kovač, *Modal collapse in Gödel's ontological proof*, [in:] M. Szatkowski (ed.), **Ontological Proofs Today**, Ontos Verlag, Heusenstamm (2012), pp. 50–323.
- [26] E. J. Lowe, *A modal version of the ontological argument*, [in:] J. P. Moreland, K. A. Sweis, C. V. Meister (eds.), **Debating Christian Theism**, chapter 4, Oxford University Press, Oxford (2013), pp. 61–71.
- [27] T. Nipkow, L. C. Paulson, M. Wenzel, *Isabelle/HOL: A Proof Assistant for Higher-Order Logic*, **Lecture Notes in Computer Science**, vol. 2283 (2002).
- [28] D. S. Scott, *Appendix B: Notes in Dana Scott's Hand*, [in:] J. H. Sobel (ed.), **Logic and Theism: Arguments for and Against Beliefs in God**, Cambridge University Press, Cambridge (1972), pp. 145–146.
- [29] J. H. Sobel, *Gödel's ontological proof*, [in:] J. J. Tomson (ed.), **On Being and Saying. Essays for Richard Cartwright**, MIT Press, Cambridge, Mass. (1987), pp. 241–261.

- [30] J. H. Sobel, **Logic and Theism: Arguments for and Against Beliefs in God**, Cambridge University Press, Cambridge (2004).
- [31] K. Świątorzecka, *Identity or equality of Gödelian God*, draft paper, private communication, May 2019.

Freie Universität Berlin  
Institute of Computer Science  
Berlin, Germany  
University of Luxembourg  
Faculty of Science, Technology, and Communication  
Luxembourg  
e-mail: [c.benzmueller@gmail.com](mailto:c.benzmueller@gmail.com)

Freie Universität Berlin  
Institute of Computer Science  
Berlin, Germany  
e-mail: [david.fuenmayor@fu-berlin.de](mailto:david.fuenmayor@fu-berlin.de)